2. An agent based simulation of credit scoring

2.1 NetLogo simple model

The NetLogo simple model is the introductive approach used to understand the case of the credit market and the effects of risk propagation in supply chains. The first model is built using NetLogo, that is a simple but powerful tool that enables the simulation of natural and social phenomena. NetLogo is a free program developed by Center for Connected Learning and Computer-Based Modeling at Northwestern University, since 1999. NetLogo has several advantages, actually it is still developed, it has an extensive how-to documentation and some good tutorials. Moreover it is provided with a Models Library that contains a large variety of demonstration models that are helpful to learn how to apply the code. It is easy of use and it allows to create intuitive models that may later be implemented in lower-level platforms.

The aim of this first model is to formalize the system of interactions between firms and to study the propagation of risk and its emerging effects. The starting point of this analysis is the consideration that every time two firms interact, through the exchange of flows of goods, information and financial resources, they influence each other's conditions, both on economic and risk side. This model try to go beyond the binary interaction and reproducing the complexity of network structure and it aims to understand contagion dynamics that can arise in industrial clusters and supply chains. In order to do so it is implemented an agent based simulation in which enterprises casually interacts and influence the economic and risk conditions of the other firms. In this simulation enterprises are organized in a network in which there are clients and suppliers. The influences between firms depend on connections that are established between agents and that are graphically represented by arrows.

The world in this model is closed, so it doesn't have a toroidal form that allows agent to move directly from one side to the opposite one. In this world the axis represents the two main variables of the firms, that are an economic factor and the risk. The first one is assigned to the x-axis, while the second one is assigned to the y-axis. The risk can have only positive values, while the economic factor can be both positive and negative, in order to better correspond to real economy. With this construction the position of every firms in the world is important because the observer can graphically and immediately understand which is the global situation of every agent. As we can see in the figure below, in the quadrant I there are firms with high risk and negative economic factor, in quadrant II there are firms with high risk and positive economic factor, in quadrant III there are firms with low risk and negative economic factor and in quadrant IV there are firm with high risk and positive economic factor.

Figure 2.1



This model uses only two kind of agents: turtles and links. The former represents firms and they are square shaped. They are very simple agents and they own, as told above, two new kind of variables that are: risk and economic factor. The latter represents the connections between firms and they are the channels through which the risk propagates in the network. The links are randomly created and directional. This second characteristic of the link is helpful to understand which is the kind of relationship between two firms. The links originate from the supplier and reach the client, and so they indicate the direction of flow of goods. The kind of relationship, in this first approach, is important because only clients can influence with their conditions the variables of their suppliers. By construction two firms can't be directly one the client of the other.

On the graphic interface there are both buttons and sliders. The setup button creates a new simulation. First of all it clears previous agents and it restore the world, then it creates new firms and their links. Firms are created with random coordinates and with random values of the risk and of the economic factor. The number of the links of each firm also is random, but the average number of link can be chosen by the observer using a slider. The Run button starts the simulation. Every firm moves towards the coordinate defined by its economic factor and by its risk. Moreover after its movement every agent has a chance, defined by the observer, to modify the values of the two main variables of its suppliers. In case that it has a bigger economic value or a bigger risk it increases the respective variables of the linked agent, while it decreases them if it has lower values. The five sliders allow the observer to change the value of some crucial variable in order to obtain different results. Sliders have different functions: "num-firms" defines the number of the firms in the simulation, "average-num-links" indicates the average number of links between firms, "factor-risk-prop" represents the value of intensity of the influence of a client firm with respect to its supplier, "factor-econ-prop" has a similar meaning but for the economic variable, and "prob-of-propagation" defines the probability that a client influences the value of its in linked firm.

2.2 NetLogo second model – model 6

The second NetLogo model develops the first one, and it has the same conceptual structure, that means the same use of the Cartesian axes (on the x axis is set an economic factor and on the y axis is set the risk) and the presence of contagion dynamics but it presents many innovations. The evolution of the model is based on two different channels, that is two different strategies to manage the risk and the economic factor. In fact the first program has several deficiencies, because it allows to model the influences of risk and of an economic factor between firms but only in a naïve way, without taking in to account the exchange of flows and the probability of default. This is why in this second model firms may go bankrupt, they can arise and exchange cash flows.

The graphic interface presents new sliders, a switch and a plot. The two buttons "Setup" and "Run" have the same use of the first model so they don't need explanations. The switch called "risk-prop" has the function to separate the two influence effects, the one of the risk and the one of the economic factor, because if it is off the risk of each firm doesn't have any repercussion on others. Even in this model the sliders allow the observer to change the value of some crucial variables: "num-firms" defines the number of the firms, "average-num-links" indicates the average number of links between firms, "prob-new-firms" represents the probability of the creation of new firms, "prob-of-propagation" defines the probability that a client influences the risk of its in-linked firm, "factor-risk-migration" is the probability that a firm changes its risk level and "factor-econ-prop" represents the probability that firms change their economic situation, due to costs, revenues or financial flows. The plot shows the variation of the number of firms in time and so it helps the user to better understand the evolution of the simulation and the consequences of the mutation in the main variables. The world has a new shape, it is rectangular with 21 patches on the y axis, that represents the different classes of risk of a rating system. On the x axis is represented the cash situation that can vary between -16 and +16.

The biggest changes concern the part of procedures, because the old ones are upgraded and they are integrated with new ones. Agents, that in NetLogo have the default name of turtles, own three additional variables: "cash", "risk-asim" and "risk". The latter is the same of the first model, "cash" is a better specification of the previous economic factor, while "risk-asim" is crucial to put in place the risk migration and the risk influences. This last variable defines the probability that a firm can change its risk level: if it is bigger then 0.5 it has more likelihood to worsen it creditworthiness, on the contrary if it is lower then 0.5 it is more likely that it can improve it . It is initially set equal to 0.5 and it is bounded by two threshold: it can't be lower than 0.2 or higher than 0.8.

$$0.2 \leq risk - asim \leq 0.8$$

These limitations try to balance the simulation and to insure that a firm can always improve or worsen its risk level, even if it is not likely.

Overall there are seven procedures of which the main ones are called "setup" and "go". Both of them are linked with the two buttons of the graphic interface. The first one is used to create a new simulation: it destroys previous agents and it restore the world, and then it creates new firms and their links. Firms are created with a square shape, with “risk-asim” equal to 0.5 and with random coordinates.

```
to setup-turtles
  set-default-shape turtles "square"
  create-turtles num-firms [
    set risk-asim 50
    set cash ((random (max-pxcor * 2)) - ( max-pxcor ))
    set risk random max-pycor
    setxy cash risk
  ]
end
```

The number of the links is also random and it can be changed by the user, modifying the proper variable.

```
to setup-links
  ask turtles [
    let choice ( n-of  (random average-num-links)
      ( other turtles with [not in-link-neighbor? myself ] ) )
    if choice != nobody [ create-links-to choice ]
  ]
end
```

The “go” procedure, that is stopped when there are no turtles left in the simulation, is subdivided in three parts, referred to the movement of turtles, the creation of new firms and the update of the plot. The procedure “move-turtles” is the most complex one and it contains all the innovation of this model.

The risk is manage in a new way using a credit migration framework. Credit migration matrices are transition tables which contains the probabilities that a firm maintains or changes its risk level within the single one-year period for different classes of credit ratings. The sum of the cells in each row is 1.0 (100%) and each cell gives the probability of a migration to another rating. For examples, table 2.1, an AA rated firm is 90.75% likely to remain in the same class risk during a single period, it is 0.64% likely to upgrade to AAA and it is 7.81% likely to downgrade to A.

Table 2.1

		AAA	AA	A	BBB	BB	B	CCC	D
		***T* +1**							
T	**AAA**	**92.29**	6.96	0.54	0.14	0.06	0.00	0.00	0.000
	AA	0.64	**90.75**	7.81	0.61	0.07	0.09	0.02	0.010
	A	0.05	2.09	**91.38**	5.77	0.45	0.17	0.03	0.051
	BBB	0.03	0.20	4.23	**89.33**	4.74	0.86	0.23	0.376
	BB	0.03	0.08	0.39	5.68	**83.10**	8.12	1.14	1.464
	B	0.00	0.08	0.26	0.36	5.44	**82.33**	4.87	6.663
	CCC	0.10	0.00	0.29	0.57	1.52	10.84	**52.66**	34.030
	D	0.00	0.00	0.00	0.00	0.00	0.00	0.00	**100**

Figure 1: One-year credit migration matrix using S&P rating histories, 1981-2003. Estimation method is cohort. All values in percentage points.

Source: Schuermann (2006)

In the model there are 21 risk levels, that are represented by 21 patches on the y axis and every time a firm reaches the highest class of risk it goes bankrupt.

```
if random 100 < factor-risk-migration [
  ifelse random 100 < risk-asim [
     ifelse risk + 1 <= max-pycor [ set risk risk + 1 ] [
       die ] ] [
     if risk - 1 >= min-pycor [ set risk risk - 1 ] ] ]
```

where:

- "risk" is the risk level;
- "factor-risk-migration" is the probability that a firm remains in the same class;
- "risk-asim" is the variable that influence the direction of the grade movement.

Moreover with a certain likelihood, "prob-of-propagation", client firms can influence the risk of their suppliers. Every time a client has a bigger risk with respect to its supplier it increases by 1 unit the value of the "risk-asim" variable, that means it increases the probability of a future downgrade movement of its supplier. On the contrary if it has a smaller risk than its supplier it increases the supplier probability of a future upgrade movement.

```
if random 100 < prob-of-propagation and risk-prop = true  [
  if any? linked-turtles [
    let client-risk risk
    ask linked-turtles [
      if client-risk > risk and risk-asim + 1 <= 80 [ set risk-asim risk-asim + 1 ]
      if client-risk < risk and risk-asim - 1 >= 20 [ set risk-asim risk-asim - 1 ]
```

Also the economic factor presents important changes due to the presence of three type of economic variables: revenues, costs and payments.

- Costs affect all the type of firms, they depend on the number of out-links but they are random. Moreover the user can change the probability of costs using the slider “factor-econ-prop”.
- Revenues affect only final producers, this means firms that don’t have any out-links. Final producers are the firms which are directly linked to the market of goods and that receive money from the final clients. Revenues depend on the number of in-links and they are random too. The user can change the probability of revenues using the slider “factor-econ-prop”.
- Payments consist in cash flows that are exchanged by firms, and they move in the opposite direction of the one of the arrow of links. Payments depend on the number of in-links and the their existence is random, the user can change it with the slider “factor-econ-prop”. The total amount of payments for each firm is random, but the total value is equally dived between all its suppliers. In figure 2.2 there is an example of a supplier-client relationship between a bland firm and a final producer. In this case the bland firm is the supplier and the final producer is the client, blue arrows represent inflows while red arrows represent outflows.

Figure 2.2



Pattern of the input and output flows between two firms

The “turtles-birth” procedure manages the creation of new firms, using the same settings of the starting procedure, that means the same shape and random coordinates. In every step of the simulation is created one new turtle with a certain probability that can be changed by the user moving the “prob-new-firm” slider.
The “update-plot” procedure is necessary to update the plot on the graphic interface. The procedure identifies the right plot calling its name, then it chooses the proper pen and finally it defines the variable of interest, through the command “count turtles”.

2.3 NetLogo advanced model – dimension and economic situation

This model is the third developed and it makes the most of previous stages introducing the credit system. In fact this model purpose is to manage and to analyze banks credit policies. As in previous models the space has a specific meaning: on the x axis there is an economic factor and on the y axis there is the risk (more details are available in paragraph 2.1 NetLogo simple model)
The introduction of the lending activity allows to examine the effects for the whole system of different credit policies, that can be more bank-oriented or more system-oriented. The former tries to maximize the bank profitability, while the latter is more concerned about the common good. Through the decision of the credit risk that the bank decides to bare, it is possible to observe the consequences on the number of

firms, of the number of defaults and of repealed loans and the total of occurred losses. The model allows to change the lending policies thanks to two variable, that are the minimum initial level of credit worthiness of the borrower and the risk level necessary to maintain the loan. Using this two thresholds is possible to modify the economic cycles and to manage the economic crises generated by low level of credit availability, that is the amount of money that can be borrowed at a given time. Moreover is possible to analyze what happen to the economic system when the bank decide to penalize or to favor the firms on the basis of their dimension. This mechanism allow the user to evaluate the potential negative consequences of the Basel accord which tends to penalize SMEs.

The model presents some changes of the risk structure: it is introduced a new condition of default, it is set a link between the economic and the risk variables and moreover there are new default consequences. The previous models allow one way of default, considering only the risk side, that is the worsening of the creditworthiness of a firm that already has the highest risk (the 21th level). In this model a firm can go bankrupt in any risk level with default probabilities that depend on their ratings, using a similar approach presented in table 2.2.

Table 2.2

AAA	0,01%	BBB-	0,32%
AA+	0,02%	BB+	0,53%
AA	0,03%	BB	0,93%
AA-	0,04%	BB-	1,57%
A+	0,05%	B+	2,64%
A	0,07%	B	4,46%
A-	0,09%	B-	7,52%
BBB+	0,13%	CCC	13,00%
BBB	0,18%	CC	26,00%

S&P expected default rate in 1 year.

Every time a firm fails it influences all its clients and its suppliers, this mechanism is introduced to analyze the risk contagion dynamics in the network of enterprises. For clients is increased the probability to worsen their risk class while for suppliers is directly increased their class risk. The effect towards the supplier is bigger than the one towards the clients, because for a real firm it is easier to bear the cost of a supplier default.

Figure 2.3



The pattern of the consequences of the link between risk and cash

Moreover it is introduced a link between the risk and the economic factor, that depends on the firm position in the world. In order to highlight the mechanism for the user, the world is divided in two parts by a vertical yellow line. On the left side of there are the firms with a low cash level, while on the right side there are the firms with an high one. The model increases the “risk-asim” variable, that means incrementing the probability to worsen their creditworthiness, for the firms with a bad economic situation. On the other side, for the firms with high cash level, it decreases the “risk-asim” variable, incrementing the probability to improve their creditworthiness, as it is shown in figure 2.3.

Firms are created using the same formulation of previous models, but they own additional variables (finalproducer?, net-payment, loan?, maturity?, life) and they can be of two types. In fact besides the normal firm, the bland agent, the model presents also the final producer, that is an example of tasty agent.

Final producer firms have a direct contact to retail clients and so they have “final” revenues, while normal agents are only linked with other firms, from which they receive payments from other firms. Final producer firms can be easily identified because they have a triangular shape and a bigger size. Moreover final producer are also characterized by the firm variable “finalproducer?” bigger than zero, this value represents the number of links with the final market and as a consequence the final revenues. The starting number of firms is 50 and there is the same probability to create tasty and bland agents. Moreover each new firm has some variables with default values such “risk-asim”, ”life” and “maturity?”, in order to conform their behaviors.

```
create-turtles 50 [
  set risk-asim 50
  set life 0
  set maturity? -1
  ifelse random 99 < 50 [ set finalproducer? 0 ] [ set finalproducer? 1 + random 5 ]
  if finalproducer? > 0 [
    set size 1.35
    set shape "triangle" ]
```

Firms are connected by links, which identify the client-supplier relationship and that are the channel through which agents can influence each other. Links are created with turtles but they can change during the simulation thanks to the “links-renovation” procedure.

The central procedure of the simulation is called “move-turtles”, it is a part of the “go” procedure and it is articulated in eight main parts, that are going to be deeply analyzed:

- influence-of-position;
- risk-consequences;
- manage-flows;
- collect-info;
- contagion;
- take-credit;
- give credit;
- time.

This subdivision is really useful to understand actions and influence effects of the firms and to try to give a reasonable meaning to the simulation results.

The first phase, “influence-of-position”, regards the consequences of the cash situation on the risk level, and it put in place the link between the two crucial agents variables. As it is explained before if the cash value is greater or equal to 16, the “risk-asim” variable is decreased and it is more likely that the firm improves its risk situation, while if the cash is less than 16, the “risk-asim” variable is increased.

```
to influence-of-position
  ask turtles [
    if xcor >= 16 and risk-asim - 3 >= 10 [ set risk-asim risk-asim - 3 ]
    if xcor <  16 and risk-asim + 3 <= 90 [ set risk-asim risk-asim + 3 ] ]
```

The second phase, “risk-consequences”, contains a lot of instructions that are related to the probability of default and of the change of the risk level.

On the y axis there are 21 patches, that represents 21 risk levels and credit ratings. For each rating there is a different probability of default, as it can be seen in the example of table 2.2. The probabilities don’t follow a linear distribution and so it is applied a specific formula, based on a exponential transformation, to attach the optimal probability to each rating class (Appendix B contains a detailed explanation of the construction of the function). The following function is important because it is created using real probability of default of rating classes, and for this reason it represents an anchor to the reality.

$$\text{Pd} = 10 \wedge \left(\frac{\text{ycor} - 23.42624}{5.85656108}\right)$$

Where:

- Pd = the probability of default;
- ycor = the value of the y coordinate which represents the risk class.



Probability of default for the 21 risk class

In the case of default, the firm influences both clients and suppliers, that are respectively indentified by “in-linked-turtles” and “out-linked-turtles”. The effect on suppliers, as it can be seen in the code below, is much stronger than the one on clients.

```
if random-float 1 < ( 10 ^ (( ycor - 23.42624) / 5.85656108)) / a
  ask in-linked-turtles [
    if risk + 1 <= max-pycor [ set risk risk + 1 set ycor risk ] ]
  ask out-linked-turtles [
    if risk-asim + 1 <= 90 [ set risk-asim risk-asim + 1 ] ]
  die ]
```

At the end of each year of life the risk level of each firm can change on the basis of a migration matrix. With a probability of 92% they maintain the same creditworthiness, with a probability that depends on the “risk-asim” value they increase their risk and with a complementary value of “risk-asim” they decrease their risk.

Migration matrix				
		T + 1		
		x + 1	x	x - 1
T	x + 1	94%	risk-asim	0%
	x	1-risk-asim	94%	risk-asim
	x - 1	0%	1-risk-asim	94%

The matrix presents the probability of change or of maintenance of the risk level, which is identifies by x.

In case of default, that occurs when the risk level exceeds the maximum one, the firm influences both clients and suppliers as explained before and then it disappears.

```
let birthd life / 26
if ( int birthd ) - birthd = 0 [
  if random 100 < 92 [
    ifelse random 100 < risk-asim [
      ifelse risk + 1 <= max-pycor [ set risk risk + 1 ] [
        ask in-linked-turtles [
          if risk + 1 <= max-pycor [
            set risk risk + 1
            set ycor risk ] ]
        ask out-linked-turtles [
          if risk-asim + 1 <= 90 [ set risk-asim risk-asim + 1 ] ]
        default-consequences ] ] [
      if risk - 1 >= min-pycor [ set risk risk - 1 ] ] ]
```

The third procedure, “manage-flows”, is dedicated to the economic variables. In fact the cash of a firm can change in two ways: it can increase due to revenues or it can decrease because of costs. The latter affects every single firm, and it depends on a random function that includes the number of in-links and the “costs” variable of the interface slider. The former affects only the firms that are final producers, that means with the variable “finalproducer?” bigger than zero, and it depends on a random function that includes the number of links to the final market and the slider “revenues” variable.

```
if finalproducer? > 0 [
  let revenuess ( ( (revenues * finalproducer? ) ) + random-normal 0 2 )
  if revenuess > 0 [ set cash cash + revenuess ]
  ]
let cost ( ( ( costs * cash * 2.1 ) ) + random-normal 0 2 ) / 20
if cost > 0 [ set cash cash - cost ]
if int xcor != cash [
  facexy cash ycor forward 0.5 ] ]
```

The fourth phase, “collect-info”, is created to avoid the problems of sequentiality of payments. The latter is defined by a random function that depends on the number of in-links, that means the number of suppliers. Payments differ on the basis of the type of firm. For normal firms payments depend on a variable called “payments-amount” that can be change by the user on the graphic interface, while for final producer they depends on revenues. In order to settle the payments in a contemporary and parallel way, for each firm the inflows and the outflows are collected in a specific variable called net-payment.

```
let linked-turtles in-link-neighbors
if any? linked-turtles [
  if finalproducer? = 0 [
    let payments ( ( payments-amount * 0.1 * count my-in-links ) + random-normal 0 0.5 ) / 2
    if payments > 0 [
    set net-payment net-payment - payments
    let payment payments / count my-in-links
    ask linked-turtles [ set net-payment net-payment + payment ] ] ]
  if finalproducer? > 0 [
    let payments ( ( revenues * 0.9 * count my-in-links ) + random-normal 0 0.5 ) / 2
    if payments > 0 [
    set net-payment net-payment - payments
    let payment payments / count my-in-links
    ask linked-turtles [ set net-payment net-payment + payment ] ] ] ]
```

The fifth phase, “contagion”, concern the influences of risk and of the economic factor between firms.

With a certain likelihood, “prob-of-propagation”, client firms can influence the risk of their suppliers, which are represented by “in-link-neighbors” agentset. Every time a client has a bigger risk with respect to its supplier, the value of the “risk-asim” variable is increased by 0.5 unit, that means it increases the probability of a future downgrade movement of its supplier. On the contrary if it has a smaller risk than its supplier it increases the supplier probability of a future upgrade movement.

```
ask turtles [
  let in-linked-turtles in-link-neighbors
  if random-float 1 < prob-of-risk-propagation [
    if any? in-linked-turtles [
      let client-risk risk
      ask in-linked-turtles [
        if client-risk > risk and risk-asim + 0.5 <= 90 [ set risk-asim risk-asim + 0.5 ]
        if client-risk < risk and risk-asim - 0.5 >= 10 [ set risk-asim risk-asim - 0.5 ] ] ] ]
```

The same procedure is applied to the influences towards the clients.

```
let out-linked-turtles out-link-neighbors
if random-float 1 < prob-of-risk-propagation [
  if any? out-linked-turtles [
    let supplier-risk risk
    ask out-linked-turtles [
      if supplier-risk > risk and risk-asim + 0.5 <= 90 [ set risk-asim risk-asim + 0.5 ]
      if supplier-risk < risk and risk-asim - 0.5 >= 10 [ set risk-asim risk-asim - 0.5 ]
```

Moreover in this part of the code the payment procedure is completed, because the “net-payment” variable, that contains the sum of all the inflows and the outflows, is added to the cash of the firm and then it is restored to zero. After this operation each firm moves towards its right position on the x axis.

```
ask turtles [
  set cash cash + net-payment
  set net-payment 0
  if int xcor != cash [
    facexy cash ycor forward 0.5
    ] ]
```

The following two procedures are related to the credit services, they are “take-credit” and “give-credit”, and they contain all the instructions that allow firms to receive a loan when they need it.

In order to consider the consequences of the lending decisions on the basis of firms dimension a new risk variable is introduced. It is called perceived-risk and it is the result of the multiplication of the real risk level with a dimension coefficient, “premium-large-firms”. This coefficient, that can be modified by users on the graphic interface, vary from 0.1 to 10, and this means that the risk of a firm can be increased or reduced up to ten times. If the premium is equal to 10 then the risk of large firm is reduced and at the same time the risk of small firms is increased, and this is very important element to consider. Thanks to this mechanism we can reinforce the effects and we can evaluate the consequences of lending decision in a easier way. The premium is applied to firms considering their dimensions, and so it is important to divide them in two categories. The firms that have more in-links then the average number of links are considered large firms, on the contrary firms that have less in-links then the average are considered small firms.

```
if count my-in-links <= average-num-links [ set perceived-risk ycor * premium-large-firms ]
if count my-in-links > average-num-links [ set perceived-risk ycor / premium-large-firms ]
```

There are three main parts of the code that are related to three different situations:

- the loan repayment, when “maturity?” is equal to 0 (take-credit);
- the monitoring of the credit situation of the borrower, every time “maturity?” is bigger the 0 (give-credit);
- the granting of the loan, when “maturity?” is equal to -1 (give-credit).

The “maturity?” variable of firms indicates the number of steps left before the expiration of the loan. This variable is set equal to -1 by default every time a new firm is created and it is reduced by 1 unit at each cycle of the simulation if there is an existing loan.

A loan is granted only if three conditions are met: “maturity?” equal to -1, that means that there are no existing loan, the cash level has to be less the 16 and the perceived risk level has to be equal or less than a chosen threshold. These threshold is called “initial-risk-threshold” and it is a slider variable, that can be changed by the user. The latter can also modify the total amount of the loan, “loan?”, and also the starting maturity, “maturity?”. Using these three variables is possible to modify the lending policy of the banks obtaining different results.

```
if maturity? = -1 and perceived-risk <= initial-risk-threshold and xcor < 16 [
  set cash cash + loan-amount
  set loan? loan-amount
  set total-loans total-loans + loan-amount
  set maturity? maturity + 1 ] ]
```

The banks steadily monitor the existing loans, if “maturity?” is greater than 0, checking the evolution of the creditworthiness of the borrowers. In case that the perceived risk level of one borrower exceeds the chosen threshold, “monitoring-risk-threshold”, the loan is repealed. This action can determine the default of the firm and all the consequences of the risk diffusion.

```
if maturity? > 0 and perceived-risk > monitoring-max-risk-threshold [
  set cash cash - loan?
  if cash - loan? < 0 [ set insolvency-losses insolvency-losses - ( cash - loan? ) ]
  set loan? 0
  set maturity? 0
  if cash <= 0 [
    ask in-linked-turtles [
      if risk + 1 <= max-pycor [
      set risk risk + 1
      set ycor risk ] ]
    ask out-linked-turtles [
      if risk-asim + 1 <= 90 [ set risk-asim risk-asim + 1 ] ]
    set repealed-loans repealed-loans + 1
    default-consequences ] ]
```

When it is reached the expiration date, that means that the “maturity?” is equal to 0, the firm return the loan amount to the bank. Also this action can determine the firm default and the risk diffusion as explained before.

```
if maturity? = 0 [
  if cash >= 0 and cash - loan? < 0 [ set insolvency-losses insolvency-losses - ( cash - loan? ) ]
  if cash < 0 [  set insolvency-losses insolvency-losses + loan? ]
  set cash cash - loan?
  set loan? 0
  if cash <= 0 [
    set insolvencies insolvencies + 1
    ask in-linked-turtles [
      if risk + 1 <= max-pycor [
        set risk risk + 1
        set ycor risk ] ]
    ask out-linked-turtles [
      if risk-asim + 1 <= 90 [ set risk-asim risk-asim + 1 ] ]
    default-consequences ] ] ]
```

The last procedure is called “time” and it has the function of measure the passing of time. It is short and simple, but it has an important underlying hypothesis. In fact it is assumed that to every tick, that means to every cycle of the simulation, corresponds two weeks of real life, considering that on average firms carry out payments every fifteen days.

```
to time
  tick
  ask turtles [
    if maturity? >= 0 [ set maturity? maturity? - 1 ]
    set life life + 1 ]
end
```

The model presents several monitoring tools, that are represented by plots and monitors.

The “Number of firms” plot indicates the evolution of the number of firms during the simulation.



The “Defaults” plot shows the total number of defaults that have occurred.



The “Default/firms per tick” plot presents the percentage of the ratio of the number of defaults to the number of firms for every cycle of the simulation.



In order to have a global view the interface of the advanced model is represented in the following image.



2.4 NetLogo advanced model results

The result of the simulations are considered in two different economic situations, that are stability or instability. The user can easily switch from one to the other using the specific chooser of the graphic interface called “economic_stability”.

The first situation is characterized by a low level of default, due to the high resistance of firms, and for this reason the bank doesn’t play such a crucial role in the economy. While in the second situation there is a high level of default, the average life of firms is reduced and the bank become really important.

All the simulations considered in this documentation use the same starting variables, that can be observed in the following image, and the only differences concern the economic stability and the “premium-large-firms”. The latter is set first equal to 10.0 and then equal to 0.1, in order to consider the extreme cases of discrimination of firms on the basis of their dimension. The following four cases of simulation consider 10000 cycles that correspond to about 380 years.



Case 1 – Economic Stability and premium for large firms set equal to 10.0









Case 2 – Economic Stability and premium for large firms set equal to 0.1









In a situation of stability there are no significant differences between the penalization of small firms or large firms. The average number of firms is slightly bigger in the case in which large firms are favored, but at the same time the total amount of loan is smaller. Moreover you can notice that in case 1 the average life of firm is bigger and so if you favor large firms you can obtain better systemic results.

Case 3 – Economic Instability and premium for large firms set equal to 10.0









Case 4 – Economic Instability and premium for large firms set equal to 0.1







In a situation of instability there are results that are similar to the first two cases.
If the bank favor large firms you can notice a bigger average number of firms and a bigger average life. Moreover in the fourth case the simulation has stopped before 10000 cycles because of the default of all firms. This last dramatic consequence inevitably occurs every time the bank favor SME and so penalize large firms.
In conclusion, the results of the model, which say that disastrous systematic consequences occur only if the bank penalize large companies, would reduce concerns about the negative consequences of the Basel 2 which tends to penalize SMEs.